
Abstract
Substitution is a common and popular approach to implementing name binding in definitional
interpreters. A common pitfall of implementing substitution functions is variable capture. The
traditional approach to avoiding variable capture is to rename variables. However, traditional
renaming makes for an inefficient interpretation strategy. Furthermore, for applications where
partially-interpreted terms are user facing it can be confusing if names in uninterpreted parts of
the program have been changed. In this paper we explore two techniques for implementing capture
avoiding substitution in definitional interpreters to avoid renaming.

1 Introduction

Following Reynolds [22], a definitional interpreter is an important and frequently used method
of defining a programming language, by giving an interpreter for the language that is written
in a second, hopefully better understood language. The method is widely used both for
programming language research [3, 4, 13, 19, 23] and teaching [15, 20, 24]. A commonly used
approach to defining name binding in such interpreters is substitution. A key stumbling block
when implementing substitution is how to deal with name capture. The issue is illustrated
by the following untyped λ term:

(λf. λy. (f 1) + y) (λz. y︸︷︷︸
free variable

) 2 (1)

This term does not evaluate to a number value because y is a free variable; i.e., it is not
bound by an enclosing λ term. However, using a naïve, non capture avoiding substitution
strategy to normalize the term would cause f to be substituted to yield an interpreter state
corresponding to the following (wrong) intermediate term (λy. ((λz. y) 1) + y) 2 where the
red y is captured; that is, it is no longer a free variable.

Following, e.g., Curry and Feys [12], Plotkin [21], or Barendregt [5], the common technique
to avoid such name capture is to rename variables either before or during substitution (a
process known as α-conversion [11]). For example, by renaming the λ bound variable y to r,
we can correctly reduce term (1) to (λr. ((λz. y) 1) + r) 2.

While a renaming based substitution strategy provides a well behaved and versatile
approach to avoiding name capture, it has some trade-offs. For example, since renaming
typically works by traversing terms, interpreters that rename at run time are typically slow.
Furthermore, renaming gives intermediate terms whose names differ from the names in source
programs. For applications where intermediate terms are user facing (e.g., in error messages,
or in systems based on rewriting) this can be confusing. For this reason, interpreters often

2:2 Renamingless Capture-Avoiding Substitution for Definitional Interpreters

use alternative techniques for (lazy) capture avoiding substitution, such as closures [16], De
Bruijn indices [14], explicit substitutions [1], locally nameless [9]. However, traditional named
variable substitution is sometimes preferred because of its simple and direct nature.

This paper explores named substitution strategies that do not rename variables. We
explore two such strategies. The first technique we explore is a technique that Eelco Visser
and I were using to teach students about static scoping, by having students implement
definitional interpreters. To this end, we used a simple renamingless substitution strategy
which (for applications that do not perform evaluation under binders) is capture avoiding.
The idea is to delimit and never substitute into those terms in abstract syntax trees (ASTs)
where all substitutions that were supposed to be applied to the term, have been applied;
e.g., terms that have been computed to normal form. For example, using ⌊ and ⌋ for this
delimiter, an intermediate reduct of the term labeled (1) above is (λy. ( ⌊(λz. y)⌋ 1) + y) 2.
Here the delimited highlighted term is closed under substitution, such that the substitution
of y for 2 is not propagated past the delimiter; i.e., using ; to denote step-wise evaluation:

(λf. λy. (f 1) + y) (λz. y) 2

; (λy. ( ⌊(λz. y)⌋ 1) + y) 2

; ( ⌊(λz. y)⌋ 1) + 2

; ((λz. y) 1) + 2
; y + 2

The result term computed by these reduction steps is equivalent to using a renaming based
substitution function. However, the renamingless substitution strategy we used does not
rename variables (and so preserves the names of bound variables in programs), is simple
to implement, and is more efficient than interpreters that rename variables at run time. A
limitation of the renamingless substitution strategy is that it is not well-behaved for reduction
strategies that perform evaluation under binders. That is, the strategy assumes that we treat
any function expression λx.e as a value, such that the expression e is only ever evaluated if
we apply the function. While this limits the applicability of the strategy, many operational
semantics of λs adhere to this restriction [16, 18, 20, 22], including the ones we considered in
the course we taught together. I never had the chance to discuss the novelty of the technique
with Eelco. However, the technique we used in the course does not seem widely known or
used. In this paper we explain and explore the technique and its limitations.

The second technique for capture-avoiding named substitution that we explore is an
existing technique by Berkling and Fehr [7] which we were made aware of by a reviewer of a
previous version of this paper. The technique has similar benefits as the technique we used
in our course: it does not rename variables and is more efficient than interpreters that do
renaming at run time. Furthermore, the technique does not make assumptions on behalf
of interpretation strategy, and it supports evaluation under binders. On the other hand,
Berkling and Fehr’s substitution technique is more involved to implement and is a little less
efficient than the renamingless substitution strategy that Eelco and I used in our course.

The renamingless techniques we consider in this paper are not new (at least the second
technique is not; we do not expect that the first one is either, though we have not found it in
the literature). But we believe they deserve to be more widely known. Our contributions are:

We describe (§ 2) a simple, renamingless substitution technique for languages with open
terms where evaluation does not happen under binders. The meta-theory of this technique
is left for future work. We discuss and illustrate known limitations in terms of examples.
We describe (§ 3) an existing and more general technique [7] which has similar benefits
and does not suffer from the same limitations. However, its implementation is a little
more involved to implement than the simple renamingless substitution strategy in § 2,
and it is a little less efficient.

This paper is a literate Haskell document, available at https://github.com/casperbp/
renamingless-capture-avoiding, and is structured as follows. § 2 describes a simple
renamingless capture avoiding substitution strategy and its known limitations and § 3
describes Berkling-Fehr substitution which has similar benefits and fewer limitations but is
less simple to implement. § 4 discusses related work and § 5 concludes.

2 Renamingless Capture-Avoiding Substitution

We present a simple technique for capture avoiding substitution, which avoids the need to
rename bound variables. To demonstrate that the technique is about as simple to implement
as substitution for closed terms (i.e., terms with no free variables, for which variable capture
is not a problem), we first implement a standard substitution-based definitional interpreter
for a language with closed, call-by-value λ expressions.

2.1 Interpreting Closed Expressions

Below left is a data type for the abstract syntax of a language with λs, variables, applications,
and numbers. On the right is the substitution function for the language. The function binds
three parameters: (1) the variable name (String) to be substituted, (2) the expression the
name should be replaced by, and (3) the expression in which substitution happens.

data Expr0

= Lam0 String Expr0

| Var0 String
| App0 Expr0 Expr0

| Num0 Int

subst0 :: String → Expr0 → Expr0 → Expr0

subst0 x s (Lam0 y e) | x ≡ y = Lam0 y e
| otherwise = Lam0 y (subst0 x s e)

subst0 x s (Var0 y) | x ≡ y = s
| otherwise = Var0 y

subst0 x s (App0 e1 e2) = App0 (subst0 x s e1) (subst0 x s e2)
subst0 (Num0 z) = Num0 z

The main interesting case is the case for Lam0. There are two sub-cases, declared using
guards (the Boolean expressions after the vertical bar). The first sub-case is when the
variable being substituted matches the bound variable (x ≡ y). Since the inner variable
shadows the outer, the substitution is not propagated into the body. In the other case
(otherwise), the substitution is propagated. This other case relies on an implicit assumption
that the expression being substituted by x does not have y as a free variable. If we violate
this assumption, the substitution function and interpreter interp0 on the left below is not
going to be capture avoiding. Below right is an example invocation of the interpreter.

interp0 :: Expr0 → Expr0

interp0 (Lam0 x e) = Lam0 x e
interp0 (Var0 ) = error "Free variable"
interp0 (App0 e1 e2) = case interp0 e1 of

Lam0 x e → interp0 (subst0 x (interp0 e2) e)
→ error "Bad application"

interp0 (Num0 z) = Num0 z

> interp0 (App0 (Lam0 "x" (Var0 "x"))
(Num0 42))

Num0 42

2.2 Intermezzo: Capture-Avoiding Substitution Using Renaming
The substitution function subst0 relies on an implicit assumption that expressions are closed;
i.e., do not contain free variables. If we want to support open expressions (i.e., expressions
that may contain free variables), we must take care to avoid variable capture. A traditional
approach [21] is to rename variables during interpretation, as implemented by the function
subst01 whose cases are the same as subst0, except for the Lam0 case:

subst01 x s (Lam0 y e) | x ≡ y = Lam0 y e
| otherwise = let z = fresh x y s e

in Lam0 z ( subst01 x s (subst01 y (Var0 z) e) )

Here fresh x y s e is a function that returns a fresh identifier if x ̸∈ FV (e) or y ̸∈ FV (s), or
returns y otherwise. While this renaming based substitution strategy provides a relatively
conceptually straightforward solution to the name capture problem, it requires an approach
to generating fresh variables, and, since it performs two recursive calls to subst01, it is
inherently less efficient than the substitution function from § 2.1 – even in a lazy language
like Haskell. Furthermore, depending on how fresh is implemented, the interpreter may
not preserve the names of λ-bound variables. In the next section we introduce an simple
alternative substitution strategy which does not rename or generate fresh variables, and
which has similar efficiency as substitution for closed expressions.

2.3 Interpreting Open Expressions with Renamingless Substitution
Let us revisit the interpretation function interp0 from § 2.1. Because our interpreter
eagerly applies substitutions whenever it can, and because evaluation always happens at the
top-level, never under binders, we know the following. Whenever the interpreter reaches an
application expression e1 e2, we know that any variable that occurs free in e2 corresponds
to a variable that was free to begin with. We can exploit this knowledge in our interpreter
and substitution function. To this end, we introduce a dedicated expression form (the
highlighted Clo1 constructor below) which delimits expressions that have been closed under
substitutions such that we never propagate substitutions past this closure delimiter:

data Expr1

= Lam1 String Expr1

| Var1 String
| App1 Expr1 Expr1

| Num1 Int
| Clo1 Expr1

subst1 :: String → Expr1 → Expr1 → Expr1

subst1 x s (Lam1 y e) | x ≡ y = Lam1 y e
| otherwise = Lam1 y (subst1 x s e)

subst1 x s (Var1 y) | x ≡ y = s
| otherwise = Var1 y

subst1 x s (App1 e1 e2) = App1 (subst1 x s e1) (subst1 x s e2)
subst1 (Num1 z) = Num1 z
subst1 ( Clo1 e) = Clo1 e

Here subst1 does not propagate substitutions into expressions delimited by Clo1 . The
interpretation function interp1 uses Clo1 to close expressions before substituting (in the
App1 case), thereby avoiding name capture:

interp1 :: Expr1 → Expr1

interp1 (Lam1 x e) = Lam1 x e
interp1 (Var1 x) = Var1 x
interp1 (App1 e1 e2) = case interp1 e1 of



C. Bach Poulsen 2:5

Lam1 x e → interp1 (subst1 x ( Clo1 (interp1 e2)) e)
e′

1 → App1 e′
1 (interp1 e2)

interp1 (Num1 z) = Num1 z
interp1 ( Clo1 e) = e

Whereas interp0 explicitly crashes when encountering a free variable or when attempting to
apply a non-function to a number, interp1 may return a “stuck” term in case it encounters a
free variable or an application expression that attempts to apply a value other than a function.
The last case of interp1 says that, when the interpreter encounters a closed expression, it
“unpacks” the closure. This unpacking will not cause accidental capture: interpretation never
happens under binders, so the only way the unpacked term can end up under a binder is via
substitution. However, interp1 only calls substitution on terms closed with Clo1 , thereby
undoing the unpacking to prevent accidental capture.

To illustrate how interp1 works, let us consider how to interpret ((λf. λy. f 0) (λz. y) 1).
The rewrites below informally illustrate the interpretation process, where for brevity we use
λ notation instead of the corresponding constructors in Haskell and ⌊e⌋ instead of Clo1 e:

interp1 ((λf. λy. f 0) (λz. y) 1)

≡ interp1 ((λy. ⌊(λz. y)⌋ 0) 1)

≡ interp1 ( ⌊(λz. y)⌋ 0)

≡ y

Unlike the renaming based substitution strategy discussed in § 2.2, our renamingless
substitution strategy does not require renaming or generating fresh variables. Its efficiency is
similar as substitution for closed expressions. It also preserves the names of binders. However,
the renamingless substitution strategy in subst1 and interp1 relies on an assumption that
evaluation does not happen under binders.

2.4 Limitation: Renamingless Substitution Does Not Support
Evaluation Under Binders

The renamingless substitution strategy from § 2.3 assumes that terms under a Clo1 have
been closed under all substitutions of variables bound in the context. Interpretation strategies
that evaluate under binders violate this assumption. For example, consider the interpreter
given by normalize1 whose highlighted recursive call performs evaluation under a λ binder:

normalize1 :: Expr1 → Expr1

normalize1 (Lam1 x e) = Lam1 x ( normalize1 e )
normalize1 (Var1 x) = Var1 x
normalize1 (App1 e1 e2) = case normalize1 e1 of

Lam1 x e → normalize1 (subst1 x (Clo1 (normalize1 e2)) e)
e′

1 → App1 e′
1 (normalize1 e2)

normalize1 (Num1 z) = Num1 z
normalize1 (Clo1 e) = e

Just like interp1, the normalize1 function closes off terms before substituting. However,
because normalize1 evaluates under λ binders, closures may be prematurely unpacked, which
may result in variable capture. For example, say we apply (λx. λy. x) to the free variable
y. We would expect the result of evaluating this application to contain y as a free variable.
However, using normalize1, the free variable y is captured:

normalize1 ((λx. λy. x) y)
≡ normalize1 (λy. ⌊y⌋)
≡ λy. normalize1 ⌊y⌋
≡ λy. y

The next section discusses a more general substitution strategy due to Berkling and Fehr [7]
which does not have this limitation, which does not rename variables, and which is more
efficient than the renaming based approach in § 2.2 but less efficient than the renamingless
substitution strategy discussed in § 2.3.

3 Berkling-Fehr Substitution

Motivated by how to implement a functional programming language based on Church’s
λ-calculus [10], Berkling and Fehr [7] introduced a modified λ-calculus which uses a different
kind of name binding and substitution. The key idea is to use a special operator (#) that acts
on variables to neutralize the effect of one λ binding. For example, in the term λx. λx. #x the
sub-term #x is a variable that references the outermost binding of x, whereas in λx. λy. #x

the sub-term #x is a free variable.
Berkling and Fehr’s # operator is related to De Bruijn indices [14] insofar as #nx acts

like an index that tells us to move n binders of x outwards. Indeed, if we were to restrict
programs in Berkling and Fehr’s calculus to use exactly one name, Berkling-Fehr substitution
coincides with De Bruijn substitution. However, whereas De Bruijn indices can be notoriously
difficult for humans to read (especially for beginners), Berkling-Fehr uses named variables
such that indices only appear for substitutions that would otherwise have variable capture.
This makes Berkling-Fehr variables easier to read for humans.

The definitions of shifting and substitution which we summarize in this section are taken
from the work Berkling and Fehr [7] with virtually no changes. However, the language we
implement is slightly different: they implement a modified λ-calculus with a call-by-name
semantics, whereas we implement the same call-by-value language as in § 2. Our purpose of
replicating their work is two-fold: to increase the awareness of Berkling-Fehr substitution and
its seemingly nice properties, and to facilitate comparison with the renamingless approach
we presented in § 2.3.

3.1 Interpreting Open Expressions with Berkling-Fehr Substitution
Below (left) is a syntax for λ expressions similarly to earlier, but now with Berkling-Fehr
indices (right) instead of variables, where Nat is the type of natural numbers:

data Expr2

= Lam2 String Expr2

| Var2 Index
| App2 Expr2 Expr2

| Num2 Int

data Index = I {depth :: Nat, name :: String}

Here the (record) data constructor I n x corresponds to an n-ary application of the special
# operator to the name x; i.e., #nx. We will refer to the n in I n x as the depth of an
index. As discussed above, a Berkling-Fehr index is similar to a De Bruijn index except
that whereas a De Bruijn index tells us how many scopes to move out in order to locate

a binder, a Berkling-Fehr index tells us how many scopes that bind the same name to
move out in order to locate a binder. In what follows, we will sometimes use λ notation
as informal syntactic sugar for the constructors in Haskell above. When doing so, we use
“naked” variables x as informal syntactic sugar for a variable at depth 0; i.e., Var2 (I 0 x).

To define Berkling-Fehr substitution, we need a notion of shifting. Shifting is used when
we propagate a substitution, say x 7→ e where x is a name and e is an expression, under a
binder y. To this end, a shift increments the depth of all free occurrences of y in s by one.
Such shifting guarantees that free occurrences of y in s are not accidentally captured.

shift :: Index → Expr2 → Expr2

shift i (Lam2 x e) | name i ≡ x = Lam2 x (shift (inc i) e)
| otherwise = Lam2 x (shift i e)

shift i (Var2 i ′) | name i ≡ name i ′

∧ depth i ⩽ depth i ′ = Var2 (inc i ′)
| otherwise = Var2 i ′

shift i (App2 e1 e2) = App2 (shift i e1) (shift i e2)
shift (Num2 z) = Num2 z

The shift function binds an index as its first argument. The name of this index (e.g., x)
denotes the name to be shifted. The depth of the index denotes the cut-off for the shift;
i.e., how many #’s an x must at least be prefixed by before it is a free variable reference
to x. For example, say we wish to shift all free references to x in the term λx. x (#x). We
should only shift #x, not x, since x references the locally λ bound x. For this reason, the
shift function uses a cut-off which is incremented when we move under binders by the same
name as we are trying to shift. For example:

shift x (λx. x (#x))
≡ λx. (shift (#x) x) (shift (#x) (#x))
≡ λx. x (##x)

The Berkling-Fehr substitution function subst2 applies shifting to avoid variable capture
when propagating substitutions under λ binders:

subst2 :: Index → Expr2 → Expr2 → Expr2

subst2 i s (Lam2 x e) | name i ≡ x = Lam2 x (subst2 (inc i) (shift (I 0 x) s) e)
| otherwise = Lam2 x (subst2 i (shift (I 0 x) s) e)

subst2 i s (Var2 i ′) | i ≡ i ′ = s
| otherwise = Var2 i ′

subst2 i s (App2 e1 e2) = App2 (subst2 i s e1) (subst2 i s e2)
subst2 (Num2 z) = Num2 z

To interpret an Expr2 application e1 e2, we first interpret e1 to a function λx. e, and then
substitute x in the body e, such that occurrences of x at a higher depth are left untouched.
But after we have substituted the bound occurrences of x in e, the depth of the remaining
occurrences of x in e need to be decremented. To this end, we use an unshift function which
decrements the depth of a given name, modulo a cut-off which now tells us what depth a
name has to strictly be larger than in order for it to be a free variable to be unshifted:

unshift :: Index → Expr2 → Expr2

unshift i (Lam2 x e) | name i ≡ x = Lam2 x (unshift (inc i) e)
| otherwise = Lam2 x (unshift i e)

unshift i (Var2 i ′) | name i ≡ name i ′

∧ depth i < depth i ′ = Var2 (dec i ′)
| otherwise = Var2 i ′

unshift i (App2 t1 t2 ) = App2 (unshift i t1 ) (unshift i t2 )
unshift (Num2 z) = Num2 z

Using unshift, we can now implement an interpreter that does evaluation under λs and that
uses capture-avoiding substitution:

normalize2 :: Expr2 → Expr2

normalize2 (Lam2 x e) = Lam2 x (normalize2 e)
normalize2 (Var2 i) = Var2 i
normalize2 (App2 e1 e2) = case normalize2 e1 of

Lam2 x e → unshift (I 0 x) (normalize2 (subst2 (I 0 x) (normalize2 e2) e))
e′

1 → App2 e′
1 (normalize2 e2)

normalize2 (Num2 z) = Num2 z

The problematic program from § 2.4 now yields a result with a free variable, as expected:

normalize2 ((λx. λy. x) y) ≡ λy. #y

3.2 Relation to Renamingless Substitution
On the surface, the techniques involved in Berkling-Fehr substitution and our renamingless
substitution strategy from § 2 may seem different. A common point between the two is
that they avoid renaming by strategically closing off certain variables to protect them from
substitutions from lexically closer binders, and strategically reopening those variables to
substitutions coming from lexically distant binders.

The renamingless substitution strategy achieves this by using a syntactic and rather
coarse-grained discipline which closes entire sub-branches over all possible substitutions,
similar to how closures à la Landin [16]. When the interpreter reaches a closed sub-expression,
it is re-opened. As discussed, this discipline works well for languages that do not perform
evaluation under binders. While we demonstrated the technique using a call-by-value
language in § 2, the technique is equally applicable to call-by-name interpretation. But not
for languages that perform evaluation under binders.

Berkling-Fehr substitution uses a more fine-grained approach to strategically close off
variables to protect them from substitutions from lexically closer binders, by shifting free
occurrences of variables when moving under a binder. When a binder is eliminated, terms
are unshifted. This fine-grained approach is not subject to the same limitations as the
renamingless approach from § 2.3. Indeed, in their paper, Berkling and Fehr [7] prove that
their notion of substitution and their modified λ-calculus is consistent with Church’s λ

calculus. Since shifting and unshifting requires more recursion over terms than the simpler
renamingless approach from § 2, Berkling-Fehr substitution is less efficient. However, it is
still more efficient than the renaming approach discussed in § 2.2.

As discussed, Berkling-Fehr substitution is closely related to De Bruijn indices, the main
difference being that Berkling-Fehr use names and are more readable. To work around the
readability issue with De Bruijn indices, one might also combine a named and De Bruijn
approach where variable nodes comprise both a name and a De Bruijn index. But that
leaves the question of how to disambiguate programs with ambiguous name. For example,

using this approach, how would the pretty-printed version of the Berkling-Fehr indexed
expression λx. λx. #x look? Berkling-Fehr indices strike an attractive balance between
efficiency, preserving names from source programs, and readability.

4 Related Work

In this paper we explored two techniques for capture avoiding substitution that avoids
renaming, for the purpose of implementing static name binding in languages with λs.
The topic of evaluating λ expressions has a long and rich history. Summarizing it all is
beyond the scope of this paper; for overviews see, e.g., the works of Barendregt [6] or
Cardone and Hindley [8]. We discuss a few of the papers that are most closely related to the
techniques we have described.

In their formalization of λ calculus and type theory, McKinna and Pollack [17] consider a
system that uses named substitution without renaming, for a particular notion of open terms.
They consider a syntax that distinguishes two classes of names: parameters and variables.
Variable substitution does not affect parameters, and parameter substitution does not affect
variables. Their notion of variable substitution is defined for terms that are variable-closed,
but which may be parameter-open. Thus, by encoding free variables as parameters, their
system can be used to compute with open terms. However, syntactically distinguishing free
variables this way seems to presupposes a static binding analysis. The approach we discussed
in § 2.3 does not presuppose such static analysis.

Our paper considers how to interpret open terms. There exist several calculi in the
literature for evaluating open terms. Accatolli and Guerrieri [2] gives an overview of several
of these calculi for open call-by-value, which is the class of languages that the interpreters in
§ 2 and § 3 interpret. Accatolli and Guerrieri focus on the meta-theory of these calculi. To
this end, they rely on an unspecified notion of capture-avoiding substitution. In this paper,
we explore how to implement such capture-avoiding substitution functions in interpreters in
a way that does not perform renaming.

5 Conclusion

We have discussed two techniques for implementing capture avoiding substitution in defini-
tional interpreters in a way that does not require renaming of bound variables. One of the
techniques relies on a coarse-grained but simple discipline for closing terms which works well
for interpretation strategies that do not evaluate under binders. The other technique, due to
Berkling and Fehr [7], is more fine-grained and also works for interpretation strategies that
evaluate under binders. While less expressive, the former technique is simpler to implement,
and is slightly more efficient. Neither of the two techniques seem to be widely known or at
least not widely applied. With this work, we hope to increase awareness of these techniques.

